

SENG 42273

Semantic web and Ontological Engineering

Lecture 9

1. Display names of all people born in London.

PREFIX dbo: <<http://dbpedia.org/ontology/>>

PREFIX dbr: <<http://dbpedia.org/resource/>>

PREFIX rdfs: <<http://www.w3.org/2000/01/rdf-schema#>>

PREFIX xsd: <<http://www.w3.org/2001/XMLSchema#>>

SELECT ...

WHERE {

...

}

```
PREFIX dbo: <http://dbpedia.org/ontology/>
PREFIX dbr: <http://dbpedia.org/resource/>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>
SELECT DISTINCT ?person
WHERE {
?person dbo:birthPlace dbr:London .
}
```

OR

```
SELECT DISTINCT ?name
WHERE {
?person dbo:birthPlace dbr:London .
?person rdfs:label ?name FILTER (lang(?name) = "en").
}
```

2. Display names and
dates of birth of all
people born in London.

```
PREFIX dbo: <http://dbpedia.org/ontology/>
PREFIX dbr: <http://dbpedia.org/resource/>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>
SELECT DISTINCT ?name ?birth
WHERE {
?person dbo:birthPlace dbr:London .
?person rdfs:label ?name FILTER (lang(?name) = "en").
?person dbo:birthDate ?birth .
}
```

3. Display names and
dates of birth of all
people born in London
between 1900-1950.

```
PREFIX dbo: <http://dbpedia.org/ontology/>
PREFIX dbr: <http://dbpedia.org/resource/>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>
SELECT DISTINCT ?name ?birth
WHERE {
?person dbo:birthPlace dbr:London .
?person rdfs:label ?name FILTER (lang(?name) = "en").
?person dbo:birthDate ?birth .
FILTER ((?birth > "1900-01-01"^^xsd:date) &&
(?birth < "1950-01-01"^^xsd:date)) .
}
```

4. Display names and
dates of birth and death
of all people born in
London between 1900-
1950.

```
PREFIX dbo: <http://dbpedia.org/ontology/>
PREFIX dbr: <http://dbpedia.org/resource/>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>
SELECT DISTINCT ?name ?birth ?death
WHERE {
?person dbo:birthPlace dbr:London .
?person rdfs:label ?name FILTER (lang(?name) = "en").
?person dbo:birthDate ?birth .
OPTIONAL { ?person dbo:deathDate ?death . }
FILTER ((?birth > "1900-01-01"^^xsd:date) &&
(?birth < "1950-01-01"^^xsd:date)) .
}
```

5. Display names and dates of birth and death of all people born in London between 1900-1950. Sort the results based on the birth date.

```
PREFIX dbo: <http://dbpedia.org/ontology/>
PREFIX dbr: <http://dbpedia.org/resource/>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>
SELECT DISTINCT ?name ?birth ?death
WHERE {
?person dbo:birthPlace dbr:London .
?person rdfs:label ?name FILTER (lang(?name) = "en").
?person dbo:birthDate ?birth .
OPTIONAL { ?person dbo:deathDate ?death . }
FILTER ((?birth > "1900-01-01"^^xsd:date) &&
(?birth < "1950-01-01"^^xsd:date)) .
}
ORDER BY (?birth)
```

6. Display names and
dates of birth of all
actors born in London
after year 1930 who are
still alive.

```
PREFIX foaf: <http://xmlns.com/foaf/0.1/>
PREFIX dbo: <http://dbpedia.org/ontology/>
PREFIX dbr: <http://dbpedia.org/resource/>
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>
SELECT DISTINCT ?name ?birth
WHERE {
  ?person dbo:birthPlace dbr:London .
  ?person dbo:occupation dbr:Actor .
  ?person foaf:name ?name .
  ?person dbo:birthDate ?birth
  FILTER (?birth > "1930-01-01"^^xsd:date) .
  FILTER NOT EXISTS { ?person dbo:deathDate ?death. }
}
```